*The Mathematica® Journal*

# *Soft Landing on the Moon with Mathematica*

**Marian Mureşan**

We consider the problem of landing a spacecraft on the Moon, assuming that aerodynamic and gravitational forces of bodies other than the Moon are negligible, and lateral motion can be ignored. Accordingly, the descent trajectory is vertical, and the thrust vector is tangent to the trajectory.

Because the spacecraft is near the Moon, we assume that the lunar acceleration of gravity has the constant value $g = 1.63$, that the relative velocity of the exhaust gases with respect to the spacecraft is constant, and that the mass rate $m'(t)$ is constrained by $-\mu \leqslant m'(t) \leqslant 0$, where $\mu$ is constant and gives the maximum rate of change of the mass due to burning the fuel.

## ■ Mathematical Approach to a Soft Landing

The problem is to make a soft landing on the surface of the Moon with the minimum amount of fuel.

Here is a sketch of the system immediately preceding the landing.

```
Graphics[{
  {Thick, Line[{{-8, 0}, {8, 0}}]},
  Line[{{-2, 5}, {2, 5}, {2, 7}, {-2, 7}, {-2, 5}}],
  Line[{{-1, 7}, {1, 7}, {1, 8}, {-1, 8}, {-1, 7}}],
  Line[{{-2, 5}, {-3, 4}}],
  Line[{{2, 5}, {3, 4}}],
  {Red, Line[{{0, 5}, {#, 4}}] & /@ Range[-.4, .4, .2]},
  Style[
   Text[Row[{"total mass = ", Style["m", Italic], "(",
      Style["t", Italic], ")"}], {-5, 6}], 12],
  Arrow[{{0, 6}, {0, 11}}],
  Style[
   Text[Row[{"thrust = ", "-", Style["k", Italic],
      " ", Style["m", Italic], "'(", Style["t", Italic],
      ")"}], {3, 9.5}], 12],
  {Dashed, Arrow[{{5, 0}, {5, 6}}]},
  Style[
   Text[Row[{"height = ", Style["h", Italic], "(",
      Style["t", Italic], ")"}], {7, 3}], 12],
  Arrow[{{0, 6}, {0, 2}}],
  Style[
   Text[Row[{Style["m", Italic], " ", Style["g", Italic]}],
    {1, 3}], 12],
  PointSize[.015], Point[{0, 6}]
 }]
```



Following [1, pp. 247–248] and [2], we introduce the following notation and assumptions:

- $t$ is time
- $m(t)$ is the mass of the spacecraft, which varies as fuel is burned
- $m'(t)$ is the rate of change of mass, constrained by $-\mu \leq m'(t) \leqslant 0$
- $g = 1.63$, the gravitational constant near the Moon
- $k$ is a constant, the relative velocity of the exhaust gases with respect to the spacecraft
- $T(t) = -k\, m'(t)$, the thrust
- $h(t)$ is the the height, with $h(t) \geq 0$
- $v(t) = h'(t)$, the velocity
- $u(t) = m'(t)$, the control function

Recalling our assumptions, aerodynamic forces and gravitational forces of bodies other than the Moon are negligible and lateral motion is ignored. Thus the descent trajectory is vertical and the thrust vector is perpendicular to the ground.

We also suppose that $m_0 = m(0) = M + F$, where $M$ is the mass of the spacecraft without fuel and $F$ is the initial mass of fuel; $m(t) > M$, since as we expect that the spacecraft will return to Earth, it needs some fuel for takeoff.

## ■ Equations of Motion

By Newton's second law ([3, p. 128] and [2]),

$$m(t)\, h''(t) = -g\, m(t) + T(t) = -g\, m(t) - k\, m'(t), \tag{1}$$

which can be written as a system of equations

$$h'(t) = v(t), \tag{2}$$

$$v'(t) = -g - k\,\frac{u(t)}{m(t)}, \tag{3}$$

$$m'(t) = u(t), \tag{4}$$

where $k$ is a constant. The third equation states that the loss of mass per second (the fuel burned by the jet per second) is proportional to the thrust of the jet.

## The Optimal Control Problem

Our goal is to minimize the fuel consumption, so the cost functional is

$$\Lambda(h, v, m, u) = -\int_0^b m'(t)\,dt = m(0) - m(b), \tag{5}$$

where $b$ is the first time for which

$$h(b) = v(b) = 0.$$

Thus the horizon is $[0, b]$, where $b$ remains to be determined.

In vector form, if $x(t) = (x_1(t), x_2(t), x_3(t)) = (h(t), v(t), m(t))$, then

$$x'(t) = f[x(t), u(t)],$$

and the problem can be written

$$\begin{aligned}
&x_1' = x_2,\\
&x_2' = -g - k\,x_3^{-1}\,u,\\
&x_3' = u \text{ (dynamics of the motion)},\\
&x_1(0) = h(0),\\
&x_2(0) = v_0,\\
&x_3(0) = M + F \text{ (initial constraints)},\\
&x_1(b) = 0,\ x_2(b) = 0,\ x_3(b) \geq M \text{ (final conditions)},\\
&u : [0, b] \longrightarrow [-\mu, 0] \text{ (} u \text{ integrable, bounded control)},\\
&t \in [0, b],\ b \geq 0 \text{ (finite horizon)},
\end{aligned} \tag{6}$$

and finally,

$$\Lambda(x, u) = -\int_0^b \frac{dx_3(t)}{dt}\,dt = x_3(0) - x_3(b) \text{ (the cost functional to minimize)}. \tag{7}$$

From (3) we have that

$$v'(t) = -g - k\,\frac{d}{dt}\ln(m(t)),$$

and by integration over the interval $[0, t]$, we get that

$$h'(t) = -g\,t - k\ln\left(\frac{m(t)}{m(0)}\right) + h'(0).$$

It follows that $h'(b) = 0$ if and only if

$$k\ln\left(\frac{m(b)}{m(0)}\right) = h'(0) - g\,b.$$

Solving for $m(b)$,

$$m(b) = m(0)\,e^{\frac{h'(0) - g\,b}{k}}.$$

Now we substitute this into (5) to get

$$\Lambda(u) = m(0)\left(1 - e^{\frac{h'(0) - g\,b}{k}}\right).$$

This result was published in [2].

**Theorem 1**

*Consider equation (5) with conditions (6) and (7). Then for some given* $h(0)$, $h'(0)$, $g$, *and* $k$, *the amount of fuel required to stop the spacecraft, that is, to force* $h(b) = v(b) = 0$, *is a strictly increasing monotonic function of the terminal time* $b$.

**Corollary 1**

*Minimizing the terminal time b for equation (5) with (6) and (7) is equivalent to minimizing the fuel consumption.*

From here it follows that instead of (7) we can consider the following cost functional

$$\Lambda(x, u) = \int_0^b dt = b \text{ to be minimized,} \tag{8}$$

and thus equation (6) with (7) becomes (6) with (8). This is a Mayer optimal control problem (see Chapter 4 of [4]).

## Necessary Conditions for the Mayer Problem

To avoid a lengthy discussion, we state a short version of theorem 4.2.i in [4]. Let the Mayer problem be expressed as

$$\Lambda(x, u) = g\,(a, x\,(a), b, x(b)) \text{ (cost functional),} \tag{9}$$

$$\begin{aligned} &\frac{dx}{dt} = f(t, x(t), u(t)),\ t \in [a, b] \text{ (a.e. (almost everywhere)),} \\ &\text{(differential constraint),} \\ &e[x] = g(a, x(a), b, x(b)) \in B \subset \mathbb{R}^{1+n+1+n} \text{ (boundary conditions),} \\ &(t, x(t)) \in T, t \in [a, b] \text{ (time-state constraint),} \\ &u(t) \in U(t), t \in [a, b] \text{ (control constraint).} \end{aligned} \tag{10}$$

A pair $(x(t), u(t))$, $a \le t \le b$, is said to be *admissible* or *feasible* provided $x : [a, b] \to \mathbb{R}^n$ is absolutely continuous [5], $u : [a, b] \to \mathbb{R}^m$ is measurable, and $x$ and $u$ satisfy (10). Let $\Omega$ be the class of admissible pairs $(x, u)$. The goal is to find the minimum of the cost functional $\Lambda(x, u)$ over $\Omega$, that is, to find an element $(x^*, u^*) \in \Omega$ such that $-\infty < \Lambda(x^*, u^*) \le \Lambda(x, u)$ for all $(x, u) \in \Omega$. We introduce the variables $p = (p_1, \ldots, p_n)$, called *multipliers*, and an auxiliary function $H(t, x, u, p)$, called the *Hamiltonian*, defined on $T \times U \times \mathbb{R}^n$ by

$$H(t, x, u, p) = \sum_{i=1}^{n} p_i f_i(t, x, u).$$

We define

$$M(t, x, p) = \inf_{u \in U(t)} H(t, x, u, p).$$

More assumptions are necessary:

**1.** There exists an element $(x^*, u^*) \in \Omega$ such that $-\infty < \Lambda[x^*, u^*] \leq \Lambda[x, u]$ for all $(x, u) \in \Omega$.

**2.** $T$ is closed in $\mathbb{R}^{1+n}$.

**3.** The set $S = \{(t, x, u) \mid (t, x) \in T, u \in U(t)\}$ is closed in $\mathbb{R}^{1+m+n}$.

**4.** $f \in C^1(S, \mathbb{R})$.

**5.** Notation:

$$f_{it} = \frac{\partial f_i}{\partial t}, f_{i x_j} = \frac{\partial f_i}{\partial x_j}, H_{x_j} = \frac{\partial H}{\partial x_j} = \sum_{i=1}^n p_i f_{i x_j}, H_t = \frac{\partial H}{\partial t} = \sum_{i=1}^n p_i f_{it}, H_{p_j} = \frac{\partial H}{\partial p_j} = f_j.$$

**6.** The graph $\{(t, x^*(t)) \mid a \leq t \leq b\}$ of the optimal trajectory $x^*$ belongs to the interior of $T$.

**7.** $U$ does not depend on time and is a closed set.

**8.** The end point $e(x^*) = g(a, x^*(a), b, x^*(b))$ of the optimal trajectory $x^*$ is a point of $B$, where $B$ possesses a tangent variety $B'$ (of some dimension $k$, $0 \leq k \leq 2n+2$), whose vectors are denoted by

$$h = \{\tau_1, \xi_1, \tau_2, \xi_2\}, \xi_1 = \left\{\xi_1^1, \ldots, \xi_1^n\right\}, \xi_2 = \left\{\xi_2^1, \ldots, \xi_2^n\right\},$$

or by

$$h = \{dt_1, dx_1, dt_2, dx_2\}, dx_1 = \left\{d\xi_1^1, \ldots, d\xi_1^n\right\}, dx_2 = \left\{d\xi_2^1, \ldots, d\xi_2^n\right\}.$$

**Theorem 2**

*Assume these eight hypotheses and let $(x^* u^*)$ be an optimal pair of the Mayer problem (9) and (10). Then the optimal pair $(x^*, u^*)$ necessarily has the following properties:*

*(a) There is an absolutely continuous function $p(t) = (p_1, \ldots, p_n)$ such that*

$$\frac{dp_i}{dt} = H_{x_i}[t, x^*(t), x^{*\prime}(t)], i = 1, \ldots, n, t \in [a, b]\ (a.e.).$$

*If $dg$ is not identically zero at $e[x^*]$, then $p(t)$ is never zero in $[a, b]$.*

*(b) For almost any fixed $t \in [a, b]$ (a.e.), the Hamiltonian (as a function depending only on $u$) takes its minimum value in $U$ at the optimal strategy $u^* = u^*(t)$, that is, $M[t, x(t), p(t)] = H[t, x(t), u^*(t), p(t)], t \in [a, b]$ (a.e.).*

*(c) The function* $M(t) = M(t, x(t), p(t))$ *coincides a.e. in* $[a, b]$ *with an absolutely continuous function and*

$$\frac{dM}{dt} = \frac{d}{dt} M[t, x(t), p(t)] = H_t[t, x(t), u(t), p(t)],\ t \in [a, b]\ (a.e.).$$

The Hamiltonian and the equations for the multipliers to (6) and (7) are

$$H = p_1 x_2 + p_2 \left(-g - k x_3^{-1} u\right) + p_3 u = (p_1 x_2 - p_2 g) + \left(-p_2 k x_3^{-1} + p_3\right) u,$$

$$p_1{}' = 0,\ p_2{}' = -p_1,\ p_3{}' = p_2 k x_3^{-2},$$

so that $p_1 = c_1$, $p_2 = -c_1 t + c_2$, $0 \le t \le b$, where $c_1$ and $c_2$ are constants.

For $-p_2 k x_3^{-1} + p_3 > 0$, the minimum of $H$ is attained with $u = 0$, and then

$$H = p_1 x_2 - p_2 g.$$

This corresponds to *free fall* for the spacecraft.

For $-p_2 k x_3^{-1} + p_3 < 0$, the minimum of $H$ is attained with $u = -\mu$, and then

$$H = p_1 x_2 + p_2\left(-g - k x_3^{-1} u\right) + p_3 u = (p_1 x_2 - p_2 g) - \left(-p_2 k x_3^{-1} + p_3\right) \mu.$$

Thus we find that the control function $u$ takes only extreme values 0 and $\mu$.

If on an interval $[\tau_1, \tau_2]$ we have that $u = 0$, and hence

$$x_1{}' = x_2,\ x_2{}' = -g,\ x_3{}' = 0,\ p_3{}' = 0,$$

then for $t \in [\tau_1,\ \tau_2]$, we have

$$x_3(t) = x_3(\tau_1),$$

$$x_2(t) = x_2(\tau_1) - g(t - \tau_1),$$

$$x_1(t) = x_1(\tau_1) - (1/2)\, g\, (t - \tau_1)^2 + x_2(\tau_1)\, (t - \tau_1).$$

In this case, $(x_1, x_2)$ describes an arc of a parabola of equation

$$x_1 = -0.5\, g^{-1} x_2^2 + c x_2 + d,\ \text{ with }\ x_2 < 0.$$

If on an interval $[\tau_1, \tau_2]$ we have $u = -\mu$, and hence

$$x_1{}' = x_2,\ x_2{}' = -g + k\, \mu\, x_3^{-1},\ x_3{}' = -\mu,$$

then for $t \in [\tau_1,\ \tau_2]$, we find

$$x_3(t) = x_3(\tau_1) - \mu(t - \tau_1),$$

$$x_2(t) = x_2(\tau_1) - g(t - \tau_1) - k \ln\left[1 - x_3^{-1}(\tau_1)\, \mu(t - \tau_1)\right],$$

$$x_1(t) = x_1(\tau_1) + x_2(\tau_1)\, (t - \tau_1) + k(t - \tau_1) - \\ 0.5\, g(t - \tau_1)^2 + k\, \mu^{-1} [x_3(\tau_1) + \mu(t - \tau_1)] \ln\left[1 - x_3^{-1}(\tau_1)\, \mu(t - \tau_1)\right].$$

**Theorem 3**

*If all the assumptions in the section Mathematical Approach to a Soft Landing are satisfied,* $\tau$ *is the time of ignition of the engine,* $b$ *is the time of landing, that is* $x_1(b) = x_2(b) = 0$*, and if the system of equations*

$$x_2(\tau_1) = g(b - \tau_1) - k \ln\left(1 - x_3^{-1}(\tau_1)\, \mu(b - \tau_1)\right),$$

$$x_1(\tau_1) = -(x_2(\tau_1) + k)\,(b - \tau_1) + 0.5\, g(b - \tau_1)^2 + \\ k\, \mu^{-1}[x_3(\tau_1) + \mu(b - \tau_1)] \ln\left(1 - x_3^{-1}(\tau_1)\, \mu(b - \tau_1)\right),$$

*for* $b$ *and* $\tau_1$ *is solvable, then there exists an optimal pair* $(x^*, u^*)$ *that solves the optimal control problem given by (6) and (8).*

## ■ Program for Soft Landing on the Moon

`MoonLanding` is a *Mathematica* program for a soft landing on the Moon. Here `h0` is the initial height, `v0` is the initial velocity, `mass` is the mass of the lander without fuel, `fuel` is the initial fuel, `g` is acceleration due to gravity, `k` is the relative velocity of the exhaust gases, and $\mu$ is the rate of change of the mass by burning.

The correctness of the results drastically depends on the initial values of the variables `z` and `g` that we use in solving the nonlinear system of equations in the program.

```
MoonLanding[h0_, v0_, mass_, fuel_, g_, k_, μ_] := Module[
  {m, z, b, sol, t1, t2},
  m = mass + fuel;
  sol =
   FindRoot[{v0 - g z - g (b - z) - k Log[1 - (μ / m) (b - z)] == 0,
     -0.5 g z^2 + v0 z + h0 + (k + v0 - g z) (b - z) - 0.5 g (b - z)^2 +
       (m k / μ) (1 - (μ / m) (b - z)) Log[1 - (μ / m) (b - z)] == 0},
    {{z, 130}, {b, 200}}];
  t1 = z /. sol;
  t2 = b /. sol;
  x1[t_] := -0.5 g t^2 + v0 t + h0;
  x2[t_] := v0 - g t;
  x32[t_] := m - μ (t - t1);
  x22[t_] := x2[t1] - g (t - t1) - k Log[1 - (μ / m) (t - t1)];
  x12[t_] := x1[t1] + (k + x2[t1]) (t - t1) - 0.5 g (t - t1)^2 +
    (m k / μ) (1 - (μ / m) (t - t1)) Log[1 - (μ / m) (t - t1)];

  Show[{
    ParametricPlot[{v0 - g t, x1[t]}, {t, 0, t1}],
    ParametricPlot[{x22[t], x12[t]}, {t, t1, t2}],
    Graphics[{
      {
```

```
    Arrow[{{-440, 0}, {800, 0}}],
    Arrow[{{0, -4000}, {0, h0 + h0 / 7}}]
   },
   {
    Text[Style["P", Italic, 12], {v0 + 30, h0}],
    Text[Style["A", Italic, 12], {x2[t1] - 20, x1[t1]}] ,
    Text[Style["O", Italic, 12], {20, 1500}],
    Text[Style["h", Italic, 12]_0, {40, h0}] ,
    Text[Style["v", Italic, 12]_0, {v0, -2000}],
    Text[Style["h_A", Italic, 12], {40, x1[t1]}] ,
    Text[Style["v_A", Italic, 12], {x2[t1], -2000}],
    Text[Style["h", Italic, 12], {-35, h0 + h0 / 10}] ,
    Text[Style["v", Italic, 12], {775, 2500}],

    Text[
     Style[Row[{Style["P" , Italic], " = (",
         Style["v" , Italic]_0, ", ", Style["h" , Italic]_0,
         ") ", " = (", v0, ", ", h0, ")" }], 12],
     {450, h0 - h0 / 10}],
    Text[
     Style[Row[{Style["m", Italic]_0, " = mass + fuel = ",
         m}], 12], {450, h0 - 2 h0 / 10}],
    Text[Style[Row[{Style["k", Italic], " = ", k}], 12],
     {450, h0 - 3 h0 / 10}],
    Text[Style[Row[{"μ = ", μ}], 12],
     {450, h0 - 4 h0 / 10}],
    Text[
     Style[Row[{Style["A" , Italic], " = (",
         Style["v_A" , Italic], ", ", Style["h_A" , Italic],
         ") = (", Chop@x2[t1], ", ", Chop@x1[t1], ")"}],
       12], {450, h0 - 5 h0 / 10}]
   },
   {
    PointSize[.01],
    Point[{{v0, h0}, {x2[t1], x1[t1]}, {0, 0},
       {0, h0}, {0, x1[t1]}, {v0, 0}, {x2[t1], 0}}]
   }
  }]
 }, PlotRange → {{Automatic, All}, {0, All}},
 ImageMargins → {{10, 50}, {10, 10}}, Axes → False,
 AxesOrigin → {0, 0}, AspectRatio → 1 / 1.3]
]
```

```
MoonLanding[50 000, -150, 2010.51, 719.63, 1.63, 636,
 16.5]
```




This `Manipulate` lets you vary the parameters in real time.

```
Manipulate[
 Quiet@MoonLanding[h0, v0, mass, fuel, g, k, μ],
 {{h0, 50 000, "intial height h₀"}, 5000, 60 000, 1,
  Appearance → "Labeled"},
 {{v0, -150, "initial velocity v₀"}, 0, -200, -1,
  Appearance → "Labeled"},
 {{mass, 2010.51, "mass without fuel"}, 1000, 3000,
  1, Appearance → "Labeled"},
 {{fuel, 719.63, "initial fuel"}, 200, 1500, 1,
  Appearance → "Labeled"},
 {{g, 1.63, "acceleration due to gravity g"}, 1, 5,
  .01, Appearance → "Labeled"},
 {{k, 636, "relative velocity of the exhaust gases k"},
  500, 1000, 1, Appearance → "Labeled"},
 {{μ, 16.5, "rate of change of the mass μ"}, 12, 20,
  .01, Appearance → "Labeled"},
 SaveDefinitions → True
]
```


intial height h0 50000
initial velocity v0 −150
mass without fuel 2010.51
initial fuel 719.63
acceleration due to gravity g 1.63
relative velocity of the exhaust gases k 636
rate of change of the mass μ 16.5
h
P
h0
P = (v0, h0) = (−150, 50000)
m0 = mass + fuel = 2730.14
k = 636
μ = 16.5
A = (vA, hA) = (−352.563, 18772.9
A
hA
O
v

## Acknowledgments

The author expresses his gratitude to Horia F. Pop from Babes-Bolyai University, Faculty of Mathematics and Computer Science in Cluj-Napoca, Romania, for valuable discussions.

## References

[1] D. E. Kirk, *Optimal Control Theory*, Englewood Cliffs, NJ: Prentice-Hall, Inc., 1970.

[2] J. Meditch, "On the Problem of Optimal Thrust Programming for a Lunar Soft Landing," *IEEE Transactions on Automatic Control*, **9**(4), 1964 pp. 477–484. doi: 10.1109/TAC.1964.1105758.

[3] G. Leitmann (ed.), *Optimization Techniques: With Applications to Aerospace Systems*, *Mathematics in Science and Engineering*, Vol. 5, New York: Academic Press, 1962.

[4] L. Cesari, *Optimization—Theory and Applications, Problems with Ordinary Differential Equations, Applications of Mathematics*, Vol. 17, New York: Springer, 1983.

[5] M. Mureşan, *A Concrete Approach to Classical Analysis*, New York: Springer, 2009.

Marian Mureşan, "Soft Landing on the Moon with *Mathematica*," *The Mathematica Journal*, 2012. dx.doi.org/doi:10.3888/tmj.14–13.

### About the Author

Marian Mureşan is affiliated with Babes-Bolyai University, Faculty of Mathematics and Computer Science, in Cluj-Napoca, Romania. He is interested in analysis, nonsmooth analysis, calculus of variations, and optimal control.

**Marian Mureşan**
*Babes-Bolyai University*
*Faculty of Mathematics and Computer Science*
*1, M. Kogalniceanu str., 400084, Cluj-Napoca*
*Romania*
*mmarianus24@yahoo.com*
*mmarian@math.ubbcluj.ro*